Problem:
There are n processes in a queue. Each process has namei and timei. The round-robin scheduling handles the processes in order. A round-robin scheduler gives each process a quantum (a time slot) and interrupts the process if it is not completed by then. The process is resumed and moved to the end of the queue, then the scheduler handles the next process in the queue.

For example, we have the following queue with the quantum of 100ms.

A(150) - B(80) - C(200) - D(200)

First, process A is handled for 100ms, then the process is moved to the end of the queue with the remaining time (50ms).

B(80) - C(200) - D(200) - A(50)

Next, process B is handled for 80ms. The process is completed with the time stamp of 180ms and removed from the queue.

C(200) - D(200) - A(50)

Your task is to write a program which simulates the round-robin scheduling.

Input

n q

name1 time1

name2 time2

...

namen timen

In the first line the number of processes n and the quantum q are given separated by a single space.

In the following n lines, names and times for the n processes are given. namei and timei are separated by a single space.

Output

For each process, prints its name and the time the process finished in order.

Constraints

1 ≤ n ≤ 100000

1 ≤ q ≤ 1000

1 ≤ timei ≤ 50000

1 ≤ length of namei ≤ 10

1 ≤ Sum of timei ≤ 1000000

Sample Input 1

5 100
p1 150
p2 80
p3 200
p4 350
p5 20

Sample Output 1

p2 180
p5 400
p1 450
p3 550
p4 800

Notes

Template in C

Solution in Haskell (judge verdict: Accepted)

import Control.Monad
import Data.List (foldl', unfoldr)

type Queue a = ([a], Int, [a], Int)

empty' :: Queue a
empty' = ([], 0, [], 0)

isEmpty' :: Queue a -> Bool
isEmpty' (_, 0, _, _) = True
isEmpty' _ = False

head' :: Queue a -> a
head' (x : _, _, _, _) = x
head' _ = error "the queue is empty"

check :: Queue a -> Queue a
check q@(f, fl, t, tl)
  | fl < tl = (f ++ reverse t, fl + tl, [], 0)
  | otherwise = q

tail' :: Queue a -> Queue a
tail' (_ : f, fl, t, tl) = check (f, fl - 1, t, tl)
tail' _ = error "the queue is empty"

snoc' :: Queue a -> a -> Queue a
snoc' (f, fl, t, tl) x = check (f, fl, x : t, tl + 1)

main = do
  [n, q] <- fmap (map read . words) getLine
  let readPair :: IO (String, Int)
      readPair = do
        [a, b] <- fmap words getLine
        return $ (a, read b)
  qu <- fmap (foldl' snoc' empty') $ replicateM n readPair
  let f qu ti o
        | isEmpty' qu = o
        | l <= q = let ti' = ti + l
                   in f t ti' ((n, ti') : o)
        | otherwise = let qu' = snoc' t (n, l - q)
                          ti' = ti + q
                      in f qu' ti' o
        where
          (n, l) = head' qu
          t = tail' qu
      o = reverse $ f qu 0 []
  mapM_ (\(n, t) -> putStrLn $ n ++ " " ++ show t) o